
	StarOffice OLE Automation with Delphi 


	



StarOffice could be a really good platform to create reports, mailings, graphs, etc, and replace Microsoft Office in those tasks. But the lack of samples could make it an impossible task. This article is intended to introduce you in the StarOffice OLE Automation with Delphi, and it throws a call to create a component suite similar to the Servers Palette in Delphi, but using StarOffice.

There are several flavours of StarOffice, it could be the Sun version, or you can use OpenOffice, it's your decision, but the OLE bridge may have differences and could cause your program to fail, you are encouraged to send any bug you find to the OpenOffice mailing lists to get it fixed.

First I'm going to put some interesting links:

StarOffice Users Group Portal
OpenOffice
The only documentation and samples I have found
I needed to create a Delphi component to convert any kind of document StarOffice can open (MS Word, etc) to StarOffice format, so I needed to use it, I started to write it, but didn't work, so I asked in the mailing list, here you can find the link to the discussion.

So I have decided to publish the solution and the source code for that simple component that could be your start point to create an StarOffice suite.

unit uStarOffice;

interface

uses Classes, ComObj;

type
    TStarOffice = class(TComponent)
    public
        Document: Variant;
        StarOffice: Variant;
        function Connect: boolean;
        procedure Disconnect;
        function CreateDocument: boolean;
        function loadDocument(const filename: string; visible: boolean): boolean;
        function saveDocumentAs(const filename: string): boolean;
        procedure CloseDocument;
        constructor Create(AOwner: TComponent); override;
        destructor Destroy; override;
    end;

procedure Register;

implementation

procedure Register;
begin
    RegisterComponents('System', [TStarOffice]);
end;

{$R *.dcr}

{ TStarOffice }

procedure TStarOffice.CloseDocument;
begin
    document.dispose;
end;

function TStarOffice.Connect: boolean;
begin
    if VarIsEmpty(StarOffice) then StarOffice := CreateOleObject('com.sun.star.ServiceManager');
    result := not (VarIsEmpty(StarOffice) or VarIsNull(StarOffice));
end;

constructor TStarOffice.Create(AOwner: TComponent);
begin
    inherited;
end;

function TStarOffice.CreateDocument: boolean;
var
    StarDesktop: Variant;
    LoadParams: Variant;
begin
    StarDesktop := StarOffice.createInstance('com.sun.star.frame.Desktop');

    LoadParams := VarArrayCreate([0, -1], varVariant);

    Document := StarDesktop.LoadComponentFromURL('private:factory/swriter', '_blank', 0, LoadParams);

    result := not (VarIsEmpty(Document) or VarIsNull(Document));
end;

destructor TStarOffice.Destroy;
begin
    inherited;
end;

procedure TStarOffice.Disconnect;
begin
    StarOffice := Unassigned;
end;

function TStarOffice.loadDocument(const filename: string; visible: boolean): boolean;
var
    StarDesktop: Variant;
    LoadParams: Variant;
    PropertyValue: Variant;
    CoreReflection: Variant;
begin
    StarDesktop := StarOffice.createInstance('com.sun.star.frame.Desktop');
    LoadParams := VarArrayCreate([0, 0], varVariant);
    CoreReflection := StarOffice.createInstance('com.sun.star.reflection.CoreReflection');
    CoreReflection.forName('com.sun.star.beans.PropertyValue').createObject(PropertyValue);
    PropertyValue.Name := 'Hidden';
    PropertyValue.Value := not visible;
    LoadParams[0] := PropertyValue;
    Document := StarDesktop.LoadComponentFromURL(filename, '_blank', 0, LoadParams);
    result := not (VarIsEmpty(Document) or VarIsNull(Document));
end;

function TStarOffice.saveDocumentAs(const filename: string): boolean;
var
    LoadParams: Variant;
begin
    LoadParams := VarArrayCreate([0, -1], varVariant);
    document.storeAsURL(filename, LoadParams);
    result := not (VarIsEmpty(Document) or VarIsNull(Document));
end;

end.
This component uses two variant variables to hold the connection to StarOffice and the Document, and also has several methods to perform operations, first you have to Connect, which returns true if everything is ok. Next, you can create a document with the CreateDocument method, load a new document stored on disk with the loadDocument method and then use SaveDocumentAs to store it on a new location. Also you can close it using CloseDocument.

Let's see the Connect method, it checks if there's an existing connection, and if not, creates one using CreateOleObject and com.sun.star.ServiceManager, if everything is ok, returns true. Now, you can use CreateDocument to create a new instance of a text document, you get an instance to a com.sun.star.frame.Desktop and call its method LoadComponentFromURL using the factory swrite which creates an instance of a text document, you can pass here any kind of StarOffice document, (spreadsheet, graph, etc).

Instead you can use the loadDocument method, to load any kind of file StarOffice can read, it creates also a desktop instance and then uses also the LoadComponentFromURL method of the desktop to load the document from the file you specify, also you can specify some properties, like Hidden to load the document in background, this is very useful to perform conversions because the user doesn't sees anything. 

You can store the current document using SaveDocumentAs, if you change the filename extension, the file will be automatically converted to the new format.

The Disconnect method, just assigns the Unassigned constant to the StarOffice connection, and to close the document, just call the dispose method of the document.

You can post here any questions, comments, corrections to this article, but I say it again, you are encouraged to create a Delphi component collection to deal with StarOffice, it will be really useful ;-) (if you know of any, please! tell me).

